Updates : 2018

****

Initialization:

```
Off[General::spell];
Off[General::spell1];
```

```
Clear["Global`*"]
```

****

Atwood Machine: simple 1 variable:

```
Clear["Global`*"]
```

```
f[1]=x[1][t]+x[2][t]-L1
```

$-L1 + x[1][t] + x[2][t]$

```
sol= DSolve[f[1]==0,x[2],t][[1]]
```

$\{x[2] \to \text{Function}[\{t\}, L1 - x[1][t]]\}$

We use DSolve so that we also substitute for derivatives of x[2]; Solve won' t work for this. (Try it!)

```
{x[2][t], x[2]'[t], x[2]''[t]} /. sol
```

$\{L1 - x[1][t], -x[1]'[t], -x[1]''[t]\}$

```
T= (1/2) Sum[m[i] D[x[i][t],t]^2 ,{i,1,2}] //.sol //Simplify
```

$\frac{1}{2} (m[1] + m[2]) x[1]'[t]^2$

```
V= (- g) Sum[  m[i] x[i][t] ,{i,1,2}] //.sol
```

$-g (m[2] (L1 - x[1][t]) + m[1] x[1][t])$

```
L=T-V
```

$g (m[2] (L1 - x[1][t]) + m[1] x[1][t]) + \frac{1}{2} (m[1] + m[2]) x[1]'[t]^2$

```
Clear[Q];
Q[i_]:=0
```

```
D[L,(x[1])[t]]
```

$g (m[1] - m[2])$

```
D[L,(x[1])'[t]]
```

$(m[1] + m[2]) x[1]'[t]$

```
 D[ D[L,(x[1])'[t]] ,t]
```

(m[1] + m[2]) x[1]''[t]

```
Lagrangian[i_]:=(
   D[ D[L,(x[i])'[t]] ,t]
   - D[L,x[i][t]] == Q[i]
 )
```

```
eqs= Lagrangian[1] //Simplify
```

g (m[1] - m[2]) == (m[1] + m[2]) x[1]''[t]

```
sol3= DSolve[eqs ,{x[1]},t] //First //Simplify
```

$\left\{x[1] \to \text{Function}\left[\{t\}, C[1] + t\,C[2] + \frac{g\,t^2\,(m[1] - m[2])}{2\,(m[1] + m[2])}\right]\right\}$

```
eqs //.sol3 //Simplify
```

True

```
initial = {x[1][0] == x0, x[1]'[0] == v0};
eqs2 = Join[{eqs}, initial]
```

{g (m[1] - m[2]) == (m[1] + m[2]) x[1]''[t], x[1][0] == x0, x[1]'[0] == v0}

```
sol4= DSolve[eqs2 ,{x[1]},t]//First //Simplify
```

$\left\{x[1] \to \text{Function}\left[\{t\},\left(g\,t^2\,m[1] + 2\,t\,v0\,m[1] + 2\,x0\,m[1] - g\,t^2\,m[2] + 2\,t\,v0\,m[2] + 2\,x0\,m[2]\right) / (2\,(m[1] + m[2]))\right]\right\}$

```
x[1][t] /. sol4
```

$\left(g\,t^2\,m[1] + 2\,t\,v0\,m[1] + 2\,x0\,m[1] - g\,t^2\,m[2] + 2\,t\,v0\,m[2] + 2\,x0\,m[2]\right) / (2\,(m[1] + m[2]))$

```
x[1][t] /. sol4 // Simplify
```

$\left(g\,t^2\,(m[1] - m[2]) + 2\,(t\,v0 + x0)\,(m[1] + m[2])\right) / (2\,(m[1] + m[2]))$

```
Collect[%, g]
```

$t\,v0 + x0 + \frac{g\,t^2\,(m[1] - m[2])}{2\,(m[1] + m[2])}$

```
eqs2 //.sol4 //Simplify
```

{True, True, True}

****

Atwood Machine: simple 2 variable

```
Clear["Global`*"]
```

```
T= (1/2) Sum[m[i] D[x[i][t],t]^2 ,{i,1,2}]
```

$$\frac{1}{2}\left(m[1]\,x[1]'[t]^2 + m[2]\,x[2]'[t]^2\right)$$

```
V= (- g) Sum[  m[i] x[i][t] ,{i,1,2}]
```

$$-g\,(m[1]\,x[1][t] + m[2]\,x[2][t])$$

```
f[1]=x[1][t]+x[2][t]-L1
```

$$-L1 + x[1][t] + x[2][t]$$

```
L=T-V
```

$$g\,(m[1]\,x[1][t] + m[2]\,x[2][t]) + \frac{1}{2}\left(m[1]\,x[1]'[t]^2 + m[2]\,x[2]'[t]^2\right)$$

```
Q[i_]:=Sum[ D[f[j],x[i][t]] lambda[j] ,{j,1,1}]
```

```
Table[Q[i],{i,1,2}]
```

$$\{\text{lambda}[1],\ \text{lambda}[1]\}$$

```
Lagrangian[i_]:=(
   D[ D[L,(x[i])'[t]] ,t]
   - D[L,x[i][t]] == Q[i]
 )
```

```
eq1= Table[Lagrangian[i],{i,1,2}];
eq1  //TableForm
```

$$-g\,m[1] + m[1]\,x[1]''[t] == \text{lambda}[1]$$
$$-g\,m[2] + m[2]\,x[2]''[t] == \text{lambda}[1]$$

```
eq2=D[f[1],{t,2}]==0
```

$$x[1]''[t] + x[2]''[t] == 0$$

```
sol1=Solve[eq2,(x[2])''[t],t][[1]]
```

$$\{x[2]''[t] \to -x[1]''[t]\}$$

```
eq2= eq1 //.sol1;
eq2 //TableForm
```

$$-g\,m[1] + m[1]\,x[1]''[t] == \text{lambda}[1]$$
$$-g\,m[2] - m[2]\,x[1]''[t] == \text{lambda}[1]$$

```
eq3 = Eliminate[eq2, lambda[1]]
```

$$(m[1] + m[2])\,x[1]''[t] == g\,(m[1] - m[2])$$

```
initial = {x[1][0] == x0, x[1]'[0] == v0};
eq4 = Join[{eq3}, initial]
```

$$\{(m[1] + m[2])\,x[1]''[t] == g\,(m[1] - m[2]),\ x[1][0] == x0,\ x[1]'[0] == v0\}$$

```
sol2= DSolve[eq4,{x[1]},t][[1]]
```

$\{x[1] \to \text{Function}[\{t\},\ (g\,t^2\,m[1] + 2\,t\,v0\,m[1] + 2\,x0\,m[1] - g\,t^2\,m[2] + 2\,t\,v0\,m[2] + 2\,x0\,m[2]) / (2\,(m[1] + m[2]))]\}$

```
eq5 = eq2 /. sol2
```

$\left\{-g\,m[1] + \frac{m[1]\,(2\,g\,m[1] - 2\,g\,m[2])}{2\,(m[1] + m[2])} == \text{lambda}[1],\ -g\,m[2] - \frac{m[2]\,(2\,g\,m[1] - 2\,g\,m[2])}{2\,(m[1] + m[2])} == \text{lambda}[1]\right\}$

```
Solve[eq5[[1]], lambda[1]][[1]]
```

$\left\{\text{lambda}[1] \to -\frac{2\,g\,m[1]\,m[2]}{m[1] + m[2]}\right\}$

```
x[1][t] /. sol2 // Simplify // Collect[#, g] &
```

$t\,v0 + x0 + \frac{g\,t^2\,(m[1] - m[2])}{2\,(m[1] + m[2])}$

```
x2Sol = DSolve[f[1] == 0, {x[2]}, t]
```

$\{\{x[2] \to \text{Function}[\{t\},\ L1 - x[1][t]]\}\}$

```
tmp = x[2][t] /. x2Sol /. sol2 // FullSimplify // Collect[#, t] &
```

$\left\{L1 - t\,v0 - x0 + t^2\left(\frac{g}{2} - \frac{g\,m[1]}{m[1] + m[2]}\right)\right\}$

```
MapAt[Together, tmp, {1, 4}]
```

$\left\{L1 - t\,v0 - x0 - \frac{g\,t^2\,(m[1] - m[2])}{2\,(m[1] + m[2])}\right\}$

```
x[1][t] /. sol2 // Simplify // Collect[#, g] &
```

$t\,v0 + x0 + \frac{g\,t^2\,(m[1] - m[2])}{2\,(m[1] + m[2])}$

****

Atwood Machine: 4 masses

```
Clear["Global`*"]
```

2. An Atwood machine is built as follows: A string of length L1 passes over a fixed light pulley supporting a mass m1 on one end and a pulley of mass m2 (negligible moment of inertia) on the other. Over this second pulley passes a string of length L2 supporting a mass m3 on one end and m4 on the other where m3 =/= m4. Set up Lagrange's equations for this system using the method of Lagrange multipliers. Use as coordinates the (not independent) distances of the four masses from the fixed pulley. Solve for the two forces of constraint and show that the mass m1 remains in equilibrium if:

m1 = m2 + m3 + m4 - (m4-m3)^2/(m4+m3)

```
T= (1/2) Sum[m[i] D[x[i][t],t]^2 ,{i,1,4}]
```

$$\frac{1}{2}\left(m[1]\,x[1]'[t]^2+m[2]\,x[2]'[t]^2+m[3]\,x[3]'[t]^2+m[4]\,x[4]'[t]^2\right)$$

```
V= (- g) Sum[  m[i] x[i][t] ,{i,1,4}]
```

$$-g\,(m[1]\,x[1][t]+m[2]\,x[2][t]+m[3]\,x[3][t]+m[4]\,x[4][t])$$

```
f[1]=x[1][t]+x[2][t]-L1
```

$$-L1+x[1][t]+x[2][t]$$

```
f[2]=x[3][t]+x[4][t]-2 x[2][t] -L2
```

$$-L2-2\,x[2][t]+x[3][t]+x[4][t]$$

```
L=T-V
```

$$g\,(m[1]\,x[1][t]+m[2]\,x[2][t]+m[3]\,x[3][t]+m[4]\,x[4][t])+\frac{1}{2}\left(m[1]\,x[1]'[t]^2+m[2]\,x[2]'[t]^2+m[3]\,x[3]'[t]^2+m[4]\,x[4]'[t]^2\right)$$

```
Q[i_]:=Sum[ D[f[j],x[i][t]] lambda[j] ,{j,1,2}]
```

```
Array[Q,4]
```

$$\{lambda[1],\ lambda[1]-2\,lambda[2],\ lambda[2],\ lambda[2]\}$$

```
Lagrangian[i_]:=(
   D[ D[L,(x[i])'[t]] ,t]
   - D[L,x[i][t]] == Q[i]
 )
```

```
eqs= Table[Lagrangian[i],{i,1,4}]
```

$$\{-g\,m[1]+m[1]\,x[1]''[t]==lambda[1],\ -g\,m[2]+m[2]\,x[2]''[t]==lambda[1]-2\,lambda[2],\ -g\,m[3]+m[3]\,x[3]''[t]==lambda[2],\ -g\,m[4]+m[4]\,x[4]''[t]==lambda[2]\}$$

```
eqs2={(x[1])''[t]            == - (x[2])''[t],
      (x[3])''[t]+(x[4])''[t]== 2 (x[2])''[t]
     };
```

```
sol=Solve[eqs2,{(x[1])''[t],(x[2])''[t]}][[1]]
```

$$\left\{x[1]''[t]\to\frac{1}{2}\,(-x[3]''[t]-x[4]''[t]),\ x[2]''[t]\to\frac{1}{2}\,(x[3]''[t]+x[4]''[t])\right\}$$

```
eqs3= eqs //.sol
```

$$\left\{-g\,m[1]+\frac{1}{2}\,m[1]\,(-x[3]''[t]-x[4]''[t])==lambda[1],\ -g\,m[2]+\frac{1}{2}\,m[2]\,(x[3]''[t]+x[4]''[t])==lambda[1]-2\,lambda[2],\ -g\,m[3]+m[3]\,x[3]''[t]==lambda[2],\ -g\,m[4]+m[4]\,x[4]''[t]==lambda[2]\right\}$$

```
sol2= DSolve[eqs3[[{3,4}]] ,{x[3],x[4]},t][[1]]
```

$$\left\{x[3] \to \text{Function}\left[\{t\}, C[1] + t\, C[2] + \frac{1}{2} t^2 \left(g + \frac{\text{lambda}[2]}{m[3]}\right)\right], x[4] \to \text{Function}\left[\{t\}, C[3] + t\, C[4] + \frac{1}{2} t^2 \left(g + \frac{\text{lambda}[2]}{m[4]}\right)\right]\right\}$$

```
sol3= Solve[(eqs3[[{1,2}]] //.sol2 //Simplify)
        ,{lambda[1],lambda[2]}][[1]] //Simplify
```

$$\{\text{lambda}[1] \to -((2\, g\, m[1]\, (4\, m[3]\, m[4] + m[2]\, (m[3] + m[4])))\, / (4\, m[3]\, m[4] + m[1]\, (m[3] + m[4]) + m[2]\, (m[3] + m[4]))), \text{lambda}[2] \to -((4\, g\, m[1]\, m[3]\, m[4])\, /\, (4\, m[3]\, m[4] + m[1]\, (m[3] + m[4]) + m[2]\, (m[3] + m[4])))\}$$

```
tmp = x[1]''[t] /. sol /. sol2 /. sol3 // FullSimplify
```

$$(g\, (-4\, m[3]\, m[4] + m[1]\, (m[3] + m[4]) - m[2]\, (m[3] + m[4])))\, / ((m[1] + m[2])\, m[3] + (m[1] + m[2] + 4\, m[3])\, m[4])$$

```
tmp /. sol7
```

0

Do again, but with (x[1])''[t]==0 s.t. m[1] is in equlibrium.

```
eqs4={(x[1])''[t]             == 0,
      (x[1])''[t]             == - (x[2])''[t],
      (x[3])''[t]+(x[4])''[t]== 2 (x[2])''[t]
     };
```

```
sol4=Solve[eqs4,{(x[1])''[t],(x[2])''[t],(x[3])''[t]}][[1]]
```

$$\{x[1]''[t] \to 0, x[2]''[t] \to 0, x[3]''[t] \to -x[4]''[t]\}$$

```
eqs5= eqs //.sol4
```

$$\{-g\, m[1] == \text{lambda}[1], -g\, m[2] == \text{lambda}[1] - 2\, \text{lambda}[2], -g\, m[3] - m[3]\, x[4]''[t] == \text{lambda}[2], -g\, m[4] + m[4]\, x[4]''[t] == \text{lambda}[2]\}$$

```
sol5= DSolve[eqs5[[{4}]] ,{x[4]},t][[1]]
```

$$\left\{x[4] \to \text{Function}\left[\{t\}, C[1] + t\, C[2] + \frac{1}{2} t^2 \left(g + \frac{\text{lambda}[2]}{m[4]}\right)\right]\right\}$$

```
(eqs5[[{1,2,3}]] //.sol5 //Simplify)
```

$$\left\{\text{lambda}[1] + g\, m[1] == 0, \text{lambda}[1] + g\, m[2] == 2\, \text{lambda}[2], \text{lambda}[2] + 2\, g\, m[3] + \frac{\text{lambda}[2]\, m[3]}{m[4]} == 0\right\}$$

```
sol6= Solve[(eqs5[[{1,2}]] //.sol4 //Simplify)
        ,{lambda[1],lambda[2]}][[1]] //Simplify
```

$$\left\{\text{lambda}[1] \to -g\, m[1], \text{lambda}[2] \to \frac{1}{2} g\, (-m[1] + m[2])\right\}$$

```
eqs6=eqs5  //.sol5 //.sol6 //ExpandAll
```

$$\left\{\text{True}, \text{True}, -2\, g\, m[3] + \frac{g\, m[1]\, m[3]}{2\, m[4]} - \frac{g\, m[2]\, m[3]}{2\, m[4]} == -\frac{1}{2}\, g\, m[1] + \frac{1}{2}\, g\, m[2], \text{True}\right\}$$

```
sol7=Solve[ eqs6[[3]],m[1] ][[1]] //Simplify
```

$$\left\{m[1] \to \frac{4\, m[3]\, m[4] + m[2]\, (m[3] + m[4])}{m[3] + m[4]}\right\}$$